
User commands

APL has always made it easy to handle programs as data, and for programs to create, read or modify other programs. Programmers commonly use this to extend the development environment by writing programs to help them manage other programs: filing, searching, analyzing and replacing expressions. SALT (Simple APL Library Toolkit) is an example.

User commands were introduced in version 12.1 to increase developers’ productivity by allowing them to make their own utilities available directly from the session without having to bring them in and out to use them.

User commands also exist in other APLs so this is not a unique feature of Dyalog but this implementation is different and based on Spice, a product only Dyalog offers.

In order to understand User commands an introduction to Spice is necessary.
Spice and the Spice command line
Spice extends your use of development tools. It allows you to execute code independently of the current workspace state. It works in conjunction with SALT. It is part of SALT’s tools.
Originally, Spice provided a separate command line in the IDE, AKA the Spice command bar. In V12.1 it can be used directly from the session thru the ] user command feature. Now you can keep your development tools separate from your application environment using it.

Spice already has tools for SALT, SVN and other tasks plugged in to it. These might be all you need. Or you can plug your own development tools into Spice and invoke them from its command line. How to do it will be shown here.
Making Spice available
Because Spice needs SALT’s utilities all you really need is SALT to be turned ON. When Dyalog is installed  SALT  is ON by default. In case it isn’t you can turn in ON thru the Options/Configure menu (with a restart of APL) or you can load the SALT workspace and use the <enableSALT> function
.

From then on, all Spice commands will be available as soon as you type, in the session, ] followed by a command name.

It is still possible to use the old Spice command line (an input area at the bottom of the screen) but you will find that it is not as practical as the ] in the session.
Using user commands 
All user commands are entered in the session starting with a right bracket, pretty much like system commands start with s right parenthesis.
To execute command xyz type  ]xyz  
To get some general help type  ]??   

To find all available commands type  ]?   

To find all the available commands in a specific folder type  ]? \folder\name
Example:

"22" for general help, "2CMD" for more specific info on command CMD

Group Neme Description
Demo Demo Run = Demo Script
Files Locate Returns where o regex string is found in files
Replace Replace occurrences of regex strings by others
Uz Props List property names and velues of “current object’
SALT Compare Compare versions
Explore Open 05 editor or Disk Explorer
List List objects or folders
Load Load en object (cless or nemespace or fn)
RemoveVersions Remove some or o1l versions
Save Save en object (cless or nemespace or fn)
Settings Chenge local and registry settings
Semples  sampleA A Semple SPICE Commend without parsing
sampled A Semple SPICE Commend with  parsing
Spice NewCnd Creete a new User commend interactively
Setup Personelize your environment (edit Spice\Setup to do so)
Uloed Load o User commend's cless in the ws
Utils in Define en ATF file's objects in the ws
inx Define en extended trensfer file's objects in the ws
out Create an ATF file from the current s
outx Creste an extended trensfer file from the current ws
Workspace Calls Returns the calling tree of e function in a scriptfile
Summary Returns o summery of the objects in a scriptfile
Xref Returns Cross-reference of the objects in a scriptfile
elign Align comments to start at a specific offset
bm Benchmerk expression
cd Chenge Directory (D0S)
display DISPLAY expression
fndiff Compares functions
fnsLike Returns o list of fns/ops metching the REGEX pattern

Ready...

[Tns [ [

CurObj: tmp (Undefined)

&:1 [Opa:0 [O7RAP [OSI:0 [OI0:1 [OML:0





To find help on a particular command type  ]?cmdname . For example, to find help on command ‘Locate’:

Command "Locate”. Syntax: 1[L] arguments: accepts switches -di
script Tocation: C:\Progrem Files (x86)\Dyalog\121U\SALT\Spice\Locate

Arg: regex string: shows where the string is found
-dir specifies @ folder different than the current workdir ol

Resdy... [tns =1

CurObj: Tocate (Undefined)  [&:1  [0DQ:0 [OTRAP [OSI:0 [OI0:1 [OML:0





The names of commands are case insensitive, so Locate and locate are the same command.

Upon hitting Enter, the line is sent to the Spice processor which then determines which command has been selected, brings in the code to run it, localized, then runs it, then cleans up.
Groups

Commands with common features can be regrouped under a single name. A group serves no other particular function. To find all the commands related to a particular group type ]?grpname
For example, to list all the commands in the utils group:

"22" for general help, "?CMD" for more specific info on command CMD

Group Neme Description

Utils in  Define an ATF file's objects in the ws
inx  Define an extended transfer file's objects in the ws
out Create an ATF file from the current ws
outx Create an extended transfer file from the current ws &

Resdy... [tns =1

CurObj: utils (Undefined)

Creating commands

A few rules must be followed. You must:
· Write a class in which the code will reside

· Put that class in a file in the Spice folder with a ‘.dyalog’ extension

· Code at least 3 specific  public shared   functions in it (described below)
A Spice class may be host to several (related) commands. Or just one.

Examples

Example #1: The TIME command 
Here is a very simple example: let’s say we want to create a user command (Spice command) that will show us the current time.
We first create a class that will handle the group of time related functions:
:Class timefns

    ⎕ML ⎕IO←1 ⍝ always set to avoid inheriting external values

    ∇ r←List
      :Access Shared Public

      r←⎕NS¨1⍴⊂''

      r.(Group Parse Name)←⊂'TimeGrp' '' 'Time'
      r[1].Desc←'Time example Script'
    ∇
    ∇ r←Run(Cmd Args)
      :Access Shared Public

      r←⎕TS[4 5 6]   ⍝ show time

    ∇
    ∇ r←Help Cmd
      :Access Shared Public

      r←'Time (no arguments)'
    ∇
:EndClass

The <List> function is used to tell Spice about the command itself. Like this, Spice is able to display a minimum of information when you type ‘]?’. This information is stored in ‘Desc’. Three more variables must be set: the command Name, the Group it belongs to and the Parsing rules. We’ll get to those rules in a bit. 

The <Help> function is used to report more detailed information when you type ‘]?time’. Since the class may harbour more than one command the function takes an argument. Here there is only one command and the argument will always be ‘time’ so we ignore it and return some help for the command ‘time’.

The <Run> function is the one executing your code for the command. It is always called with 2 arguments. Here we ignore them as all we do is call ⎕TS. Very easy.
We can write this code in a <timefns.dyalog> file using Notepad and put it in the SALT\Spice folder or write it in APL and use SALT’s  Save  command
 to put it there.

Once in the Spice folder is it available for use. All we need to do is type ]time. Et voila! The current time appears in the session as 3 numbers
.

Example #2: Another command in the same class: UTC
We may want to have another command to display the current UTC time instead of the current local time. Since this new command is related to our first ‘time’ command, we could – and should – put the new code in the same class, adding a new function <Zulu
> and modifying <Run>, <List> & <Help> accordingly. Like this:
:Class timefns

    ⎕ML ⎕IO←1

    ∇ r←List

      :Access Shared Public

      r←⎕NS¨2⍴⊂''

      r.(Group Parse)←⊂'TimeGrp' ''

      r.Name←'Time' 'UTC'

      r.Desc←'Shown local time' 'Show UTC time'

    ∇

    ∇ r←Run(Cmd Args);dt

      :Access Shared Public

      ⎕USING←'System'

      dt←DateTime.Now

      :If 'utc'≡⎕SE.Dyalog.Utils.lcase Cmd 



dt←Zulu dt


:EndIf

      r←(r⍳' ')↓r←⍕dt ⍝ remove date

    ∇

    ∇ r←Help Cmd;which

      :Access Shared Public

      which←'time' 'utc'⍳⊂⎕SE.U.lcase Cmd

      r←which⊃'Time (no arguments)' 'UTC (no arguments)'

    ∇

    ∇ r←Zulu date

     ⍝ Use .Net to retrieve UTC info

      r←TimeZone.CurrentTimeZone.ToUniversalTime date

    ∇

:EndClass

The <List> function now accounts for the ‘UTC’ command and returns a list of 2 namespaces so ‘]?’ will now return info for both commands. Same for <Help> which makes use of the <lcase> utility in []SE.Dyalog.Utils, a namespace of short utilities for use by SALT/Spice/anyone.
The <Run> function now makes use of the Cmd argument and, if it is ‘utc’, calls the <Zulu> function. It then returns the data nicely formatted, an improvement over the previous code.

Example #3: Time in Cities around the world

We could then add a new function to tell the time in Paris, another one for Toronto, etc. Each time we would have to modify the 3 shared functions above, OR, we could have a single function that takes an argument (the location) and computes the time accordingly
.  Like this:
:Class timefns

    ⎕ML ⎕IO←1

    ∇ r←List

      :Access Shared Public

      r←⎕NS¨2⍴⊂''

      r.(Group Parse)←⊂'TimeGrp' ''

      r.Name←'Time' 'UTC'

      r.Desc←'Show local time in a city' 'Show UTC time'

    ∇

    ∇ r←Run(Cmd Args);dt;offset;cities;diff

      :Access Shared Public

      ⎕USING←'System'

      dt←DateTime.Now ⋄ offset←0

      :If 'utc'≡⎕SE.Dyalog.Utils.lcase Cmd

          cities←'L.A.' 'montreal' 'copenhagen' 'sydney'

          offset←¯8 ¯5 2 10 0[cities⍳⊂⎕SE.U.lcase Args]

      :OrIf ' '∨.≠Args

          dt←Zulu dt

      :EndIf

      diff←⎕NEW TimeSpan(3↑offset)

      r←(r⍳' ')↓r←⍕dt+diff ⍝ remove date

    ∇

    ∇ r←Help Cmd;which

      :Access Shared Public

      which←'time' 'utc'⍳⊂⎕SE.U.lcase Cmd

      r←which⊃'Time [city]' 'UTC (no arguments)'

    ∇

    ∇ r←Zulu date

     ⍝ Use .Net to retrieve UTC info

      r←TimeZone.CurrentTimeZone.ToUniversalTime date

    ∇

:EndClass

Here <List> and <Help> have been updated to provide more accurate information but the main changes are in <Run> which now makes use of the Args argument. This one is used to determine if we should use the <Zulu> function and compute the offset from UTC by looking it up in the list of cities we know the time zone (offset) for.
The first argument to <Run> is always the command name (here it as called Cmd) and the second argument is whatever you entered after the command (here it is called Arg). When there are no special rules this argument will always be a string.

For example, if we enter in the Spice command line:



]time  Sydney

Cmd will contain ‘time’ and Arg will contain ‘Sydney’.
Special rules

There are times when it is easier to make a command accept variations instead of writing an entirely new command. A command switch (also known as modifier or flag or option) is an indication that the command should change its default behaviour.

For example, in SALT, the command ‘list’ is used to list files in a folder. The command accepts an argument to restrict the files to list (e.g. ‘a*’ to list only the files starting with ‘a’) and accepts also some switches (e.g. ‘-versions’ to list all the versions). Thus the command ‘]list  a*  -ver’ will only list the files starting with ‘a’ with all their versions instead of listing everything without version, which is the default.

In Spice the same thing is possible but this time you decide which switches are acceptable. When no rules are given to Spice via the Parse variable (in the <List> function) Arg  is a string and you can do whatever you wish with it. If your command is to accept switch -x then you can look for a ‘-x’ in the string and make a decision about that. It can become quite tedious to have to deal with the handling of switches every time you write a new command. Without getting into too many details let’s say that Spice takes care of that for you
. It handles switches the same way as SALT.

Let’s have a look at a more complex example.

Example #4: sample command 
Spice comes with a sample command to demonstrate the use of arguments and switches.
In file <aSample.dyalog> you will find a class with 2 commands: one which does not use the parser and one that does. 
The second command, named ‘sampleB’, uses the parser. It is similar to the ‘time/utc’ command above: it accepts one and only one argument and one switch, called TZ, which MUST be given a value. For example you could write:

Spice is unable to validate the contents of the argument but it can determine that there is only one argument. It can also ensure that TZ, if supplied, is given a value and that no other unknown switch appears.
The way to tell Spice about this is to set the ‘Parse’ variable for that command in <List> to   ‘1 -TZ=’.

The ‘1’, here, means that one and only one argument must be present and ‘-TZ=’ means “use ‘-‘ as switch delimiter, accept TZ as valid switch for the command and make sure a value is supplied (with ‘=’) whenever it is used”. Switches names are case sensitive and must be a valid APL identifier. 
If you don’t supply the number of arguments Spice won’t check its shape and you can have as many or as little (including 0) arguments as you want.

When your command is used Spice will check those conditions and if anything does not fit the Parse rules it will complain and abort execution and return the error as text result. If all goes well Spice will package the argument and switch(es) into a namespace and pass it on to <Run> as the second argument, here Arg.

Arg will contain Arguments, a list of text vectors (here only one) containing each one of the arguments and TZ which will be either the scalar number 0 if it was not specified or the string given as value if it was specified.
Let’s try to go over that again.
Here’s what the user enters in the Spice command line:


]sampleb  xyz   –TZ=123

Spice will validate this, find it is OK since there is only one argument, ‘xyz’, and that the switch TZ has been given a value, here ‘123’.

It will then call <Run> with ‘sampleB’ and a namespace containing Arguments (,⊂‘xyz’) and TZ (‘123’). The rest is up to the program to determine if this all makes sense.
Here’s another example:


]sampleB  x  y  z 

Here 3 arguments have been supplied: x, y and z and Spice won’t allow it:
Command Execution Failed: too many arguments

Another example:


]SAMPLEB  ‘x  y  z’  -TZ

Here there is only ONE argument as quotes have been used to delimit the argument of 5 characters: ‘x, space, y, space, z’ BUT the switch TZ has not been given a value so:

Command Execution Failed: value required for switch <TZ>

One more:


]Sampleb  zyx  -TT=321

Here one argument is OK but TT is not a recognized switch and:

Command Execution Failed: unknown switch: <TT>

What if we don’t supply ANY argument?



]Sampleb  -T=xx

Command Execution Failed: too few arguments

Here we supplied a proper TZ switch (Spice was able to determine that T stood for TZ) but 0 argument was not enough and therefore it complained.
As you can see Spice can be clever enough to figure out the number of arguments and which switches have been set and their values. The rules are fairly simple:
· All commands take 0 or more arguments and accept 0 or more switches

· Arguments come first, switches last

· Arguments are separated by spaces

· A special character (delimiter) identifies and precedes a switch

· Switches may be absent or present and may accept a value with the use of ‘=’

· Switches can be entered in any order

· Arguments and switch values may be surrounded by quotes to include spaces and/or switch delimiters.
Spice, after verifying that the rules are being followed correctly, will put all the arguments (the space delimited tokens) into variable ‘Arguments’ in a new namespace. It will also put in there variables of the same name as the switches. The namespace is then passed as the 2nd argument to <Run> which then runs.
There are a few more things the parser can do but this should cover most cases. See tricks below for details.
Location of commands

Spice commands are located in SALT\spice by default. You can change that by specifying a new location.

You can change the location thru the Options/Configure User Commands Tab, just remember the change won’t effective until the next restart:

You can also change the location of Spice commands immediately (no need to restart) thru the SALT command settings.
settings takes 0, 1 or 2 arguments. With 0 argument it shows the actual value of ALL settings. With 1 argument it shows the value of that particular setting. With 2 arguments it resets the value of the setting specified.

The setting to use for the Spice folder is ‘cmddir’. Thus



]settings  cmddir

will report the current Spice folders in use. The installed default is [Dyalog]\SALT\spice, which is where you find all Spice commands.

If you wish to use another folder, e.g. \my\spice\cmds you should type



]settings   cmddir   \my\spice\cmds

Note that this will change the setting for the duration of the session only. If you wish to make this permanent you should use the –permanent switch:


]settings   cmddir   \my\spice\cmds  -permanent 

More than one folder can be specified by separating the folders with semi colons (;), e.g.



]settings  cmddir  \my\spice\cmds;\my\other\goodies 

Spice will use the commands found in the folders in the order specified. If command X appears in more than one folder only the first one will be effective.

Because spaces are important in folder names you must take care NOT to introduce ANY spaces inappropriately.
When you replace the command folder by your own you effectively disable most installed commands and keep only the SALT and Spice specific commands. See below for details on those.
If you wish to ADD to the existing settings you can either retype the list of folders including the previous ones of precede your new folder with a comma to mean ADD (in front), e.g.


]settings  cmddir ,\my\spice\cmds;\my\other\goodies 

will add the 2 folders specified to any existing setting.
When you change the Spice folder, Spice waits to pick up the change. Next time you ask for ]? or a command it scans the new folder(s) specified to cache the info related to all commands: name, description, parsing rules.
Developers’ tricks
Because Spice cuts back the stack whenever an error occurs in a user command it can be difficult to debug commands as they are being created. To prevent Spice from doing so you should set SALT’s (on which Spice is based) DEBUG flag to 1, i.e.


⎕se.SALTUtils.DEBUG←1
Tracing user commands

You can trace into a user commands just like any other APL expression. Because there is a setup involved in executing a user command it can take quite a few keystrokes to get to the actual code: first the UCMD function is called then the Spice processor then your Run function. To speed up the process you can ask Spice to stop just prior to calling your Run function by adding a dash at the end of your command expressions, e.g.


]command  arguments  –

The dash will be stripped off and APL will stop on the line calling your Run function, allowing you to trace into your code.

This will only work when the DEBUG flag, as shown above, is ON.

Defaulting switch values

A switch always has a value, either 0 if not present, 1 if present without a value or a string matching the value of the switch. For example, if you set switch –X=123 then X will have the value ‘123’, not 123.

If you wish to default a switch to a specific value, you can either test its value for 0 and set it to your desired default, e.g.

:if  X=0  ⋄  X←123  ⋄  :endif

or you can use the function Switch.

Monadic Switch returns the value of the switch as if it had been requested directly except that it returns 0 for invalid switches (a VALUE error normally).

Dyadic Switch returns the value of the left argument if the switch is undefined (0) or the value of the switch if defined but with a twist: if the value of the default is numeric it assumes the value of the switch should be also and will transform it into a number, e.g. if you use  –X=123  then doing
99  Switch  ‘X’ ⍝ default to 99 if undefined

will result in  ,123 , not ‘123’

Name clashing restrictions

You should avoid using the names ‘Arguments’, ‘SwD’, ‘Switch’, ‘Propagate’ and ‘Delim’ for switch names as those names are used by the Parser itself. You can still use them but they won’t be available directly. They will only be available thru function Switch, i.e. Switch ‘SwD’ will return the value of switch SwD.
Long arguments

There are times when the number of strings for an argument may vary. To tell Spice that all the strings belong together you can use quotes to surround them. For example, if the user command newid accepts 2 arguments, say full name and address you would set Parse to ‘2’ and the user would use, e.g.


]newid  ‘joe blough’  ’42 Main str’

If the command accepts name, surname and address (3 arguments) the user would not need the quotes before ‘joe’ and after ‘blough’ but would need them for the 3rd argument otherwise Spice would complain about the number of arguments.

If it is clear that the last argument will always contain whatever is left then you can tell Spice that the command is ‘long’ and that all remaining tokens (like 42, Main and str, here) should be merged with a space between them. In that case you would tell Spice of the fact by appending the letter L to the number of arguments, e.g. ‘3L’ (plus switches if any).

An example of a command requiring one compulsory long argument would be a logging command coded ‘1L’:



]log    all this text is the argument.
New commands

When you write a new command and put it in your Spice folder, Spice does not pick up the change right away. If you subsequently try to use a user command it doesn’t know about it will rescan the folders to find it (and gather the new info, if any). If you ask for help or you have modified the parsing rules Spice may not know about it yet. To have it notice the changes use ]RESET to force a reload of all commands’ new info.
SALT user commands

Because SALT is so useful in dealing with save and load of script (.dyalog) files its commands are kept available even if you remove the Spice folder from the cmddir setting. Those commands are load, save, compare, explore, list, settings and removeversions. Of course if you shadow these with your own in a folder of your choice you will effectively get rid of them but you can always have access to SALT functionality directly by using the functions in ⎕SE.SALT, e.g. ⎕SE.SALT.Load is always available.
Spice user commands

There is only one command specifically related to Spice:

uload: allows you to load in the workspace the script related to the command given as argument. This is typically used when you are developing a command in order to modify (debug) it.
There is no need for a usave command since the save is handled by the editor’s call back function all the time.
Technicalities

User commands are implemented thru a call to ⎕SE.UCMD which is given the string to the right of the ] as the right argument and a reference to calling space as the left argument. For example, if you happen to be in namespace ABC and enter the command 



]XYZ   –mySwitch=blah

APL will make a call to ⎕SE.UCMD for you like this:



ABC  ⎕SE.UCMD ‘XYZ  –mySwitch=blah’

preserving any extra spaces you put in.

 UCMD returns a result which is displayed in the session.
Of course your code can call ⎕SE.UCMD directly at any time but if you erase or change it you will effectively disable it completely.
⎕SE.UCMD is defined to simply make a call to Spice. The example above is equivalent to 


⎕SE.SALTUtils.Spice ‘XYZ  –mySwitch=blah’

Appendix A - List of user cmds supplied
You can get an idea of the commands supplied with Dyalog by doing ]?.

Commands are divided into groups. Each group is presented here along with its commands.
Group Demo

There is only one command in this group called Demo.

Command Demo

Demo takes a script (a text file) name as argument and executes each APL line in it after displaying it on the screen.

It also sets F12 to display the next line and F11 to display the previous line. This allows you to rehearse a demo comprising a series of lines you call, in sequence, by using F12.
For example, if you wish to demo how to do something special, statement by statement you could put them in file \tmp\mydemo.txt and demo it by doing



]demo \tmp\mydemo
The extension TXT will be added if no extension is present.

The first line will be shown and executed when you press Enter. F12 will show the next which will be executed when you press Enter, etc.
Group Files

This group comprises 2 commands: locate and replace
Command Locate

This command searches the current working directory for the string given as argument in SALT script files. It needs ONE long argument which is a .Net regular expression. 
It reports all the hits in each script file where found. 

To search a different directory use the switch –dir to specify the new location.

Example:



]locate  \b\w{7}\b   –dir=\tmp

will find all 7 letter words in .dyalog files in \tmp
Command Replace

This command searches the current working directory for the string given as first argument in SALT script files and replaces occurrences by the second (long) argument. It needs TWO arguments which are .Net regular expressions.

To work on a different directory use the switch –dir to specify the new location.

Example:


    ]replace Name:\s+(\w+)\s+(\w+) Name: $2, $1 -dir=\tmp
will reverse every occurrence of  2 words when they follow ‘Name:’, i.e

Name: Joe Blough

will become

Name: Blough, Joe 
in every file it finds in the directory \tmp
Group GUI

There is only one command in this group: Props
Command Props

This command will report the properties (and their values), childlist, eventlist and proplist of the event given as argument or, if none provided, the object on which the session has focus (the object whose name appears in the bottom left corner of the session log).
Group SALT

This group contains commands that are the same as the SALT functions they name: Save, Load, List, Compare, Explore, Settings and RenameVersions.

Example:



]save myClass  \tmp\classX  -ver

This will do the same as 



⎕SE.SALT.Save ‘myClass  \tmp\classX  -ver’

Group Sample

There are commands in this group used to demonstrate the use of parsing user command lines. You should have a look at the class and read the comments in it to better understand the examples.
Command sampleA

This command is an example of a command NOT using parsing, where the argument is the entire string after the command name.

Command sampleB

This command is an example of a command using parsing, where the string after the command name is parsed and turned into a namespace containing the arguments tokenized and each switch identified.

Group Spice

This group contains two commands: NewCmd,  Setup and Uload.

Command NewCmd

This command is used to create a (series of) Spice command(s). It creates a form which is used to input all the basic information about the commands contained in a Spice class: the command names, their groups, their short and long description, their switches details.

Each command’s info is entered one after another.

When finished it creates a class which you can edit and file to be used as a personal user command(s).
Command Setup

This command is used by versions prior to V12.1 to automatically initialize Spice’s command bar to the user’s preferences. 
Command Uload

This command is used to bring in the workspace the class associated with a user command. It is used when debugging a user command in a Spice class.

Example:



]uload  newcmd

#.Setup

The class Setup containing the code the for the NewCmd user command was brought it. We can now edit the class and modify the command.

Group Utils

This group contains four commands: in, out, inx and outx.

These commands are used to transfer code in and out of Dyalog APL. See the “Dyalog APL Windows Workspace Transfer.v12.1” for details.

Group Workspace

This group contains several commands used for workspace management and debugging. Some of the commands take a pattern as argument to denote a series of objects names, for example to denote all objects starting with the letter A you would use the pattern ^A.* This is not user command specific but a use of the general regular expression syntax. It requires .Net to be present. 

Some users may be more familiar with the DOS way of using patterns where the pattern is anchored to the left, ? represent a single character and * represents 0 or more characters. Often a switch –DOS will be available to use that form. What that really means is that the DOS pattern DP becomes ^DP.* with the dot replacing the ? and .* replacing *. The commands using name patterns are fnslike, varslike, nameslike, reordlocals, sizeof and align.
Also, very often the same command will accept a –date switch which specifies the date to which the argument applies. This will typically be used when functions are involved, for example when looking for functions older than a date, say 2009/1/1, you would use –date=<90101
. The century, year and month are assumed to be the current one so if using this expression in 2009 using –date=<101 would be sufficient. You can use other comparison symbols and –date=≠80506 would look for dates other than 2008/5/6. Ranges are possible too and –date=81011-90203 would look for dates from 2008/10/11 to 2009/2/3 included.
Command Calls

This command is used to find the calls made by a program in a script file. 
It takes 2 arguments: the filename where the class resides and the function to work on. With switch –details it can provide details on all the names involved such as if the name is local, global, unused, recursively called, etc.
With switch –treeview it will show the result in a treeview window instead of the session log.

Example:

     ]calls '\Dyalog APL\12.1\SALT\SALTUtils' Spice

Level 1:  →Spice

⍝ Handle KeyPress in SPICE command window
⍝ The function can also be used directly with a string
F:isChar            F:isHelp            F:isRelPath         F:lCase             F:rlb               F:splitOn1st        F:BootSpice         F:GetSpiceList      F:SpiceHELP

Level 2: Spice→isChar

…

Level 2: Spice→BootSpice

⍝ Set up Spice
⍝ In GUI environments we rig Spice fn to be called back
F:GetSpiceList      R:Spice

Level 3: BootSpice→GetSpiceList

⍝ Retrieve the list of all Spice commands
F:getEnvir     F:lCase        F:splitOn     F:ClassFolder

Level 4: GetSpiceList→ClassFolder

⍝ Produce full path by merging root and folder name
…

At each level the calling function is followed by the called function which is detailed. It list each function called preceded by either an F (for function) or an R (for recursive call). We can see at the 1st level that function Spice calls 9 other functions and at the 2nd level function isChar calls nothing but BootSpice calls 2 functions, GetSpiceList and Spice, recursively. At the 3rd level GetSpiceList calls ClassFolder and so on.
With –details each object is preceded by either F or R as above or a character meaning:

○: local

G: global

!: undefined local

↑: glocal (global localized higher on the stack)

L: label

l: unreferenced label

*: previously described in the output
Command Summary

This command produces a summary of the functions in a class in a script file. It takes a full pathname as single argument (long).
    ]summary \Program Files\Dyalog12.1\SALT\Parser.dyalog
name       scope  size  syntax 

 fixCase             24   n0f   

 if                  24   n0f   

 init       PC     4500   n1f   

 xCut               532   r2f   

 Parse      P      5748   r1f   

 Propagate  S      1220   r2f   

 Switch            1152   r2f   

Scope shows S if shared, P if public, C if constructor and D if destructor
Size is in bytes

Syntax is a 3 letter code: 

[1] n=no result, r=result 

[2] # of arguments (valence)
[3] f=function, m=monadic operator, d=dyadic operator 

Command Xref

This command returns a Cross-reference of the objects in a script file.
It produces a very crude display of all references on top against all functions to the left. At the intersection of a function and a reference is shown symbol denoting the nature of the reference in relation to the function: o means local, G mean global, F means function, L means label.

Example:



]xref  \Program Files\Dyalog\SALT\lib\rundemo

         ccfkllnpssszzzFFILNPPS 

         llieaiaa_cn...iinieaoc 

          .lybnms ri.NRllinxtsr 

          Tes eet ip.eaeetethni 

          e . . e p .sw . . . p 

          x . . . t .t. N . . t 

          t . . . . . . . . . . 

          - - - - : - - - - : - 

 Edit     . . . . ○ . . G . : G 


 Init     . . . .○: . . . .F:GG 

 Load     .○. .○. : ○GGG.F. G G

As can be seen in this report, name script is a local in function Edit. The characters dot, dash and semi colon only serve as alignment decorators and have no special meaning.
Command align

This command will align all the end of line comments in a (list of) functions to column 40 or to the column specified with the –offset switch.
The arguments are regular expression patterns for names which can be viewed as a DOS pattern if switch –dos is supplied. The –date switch can also be applied.
The result is the list of functions that were modified in column format or in )FNS format if switch –format is supplied.

Example:



]align  HTML*  -dos  -format  -offset=60

will align all end-of-line comments at column 60 for all functions starting with ‘HTML’ and display the names of all the functions it modified in )FNS format

Command bm

This command will benchmark the expression given as argument and return the average time and CPU used for executing it 100 times. This number can be changed with the switch –try=
Example:



 ]bm     a←⌹÷∘.+⍨⍳10    -try=1000

 a←⌹÷∘.+⍨⍳10  0.062 0.061
You should assign the result of each expression if you want to avoid having it appear 100s of times on your screen.

Command cd

This command will change directory in your OS. It reports the previous directory.
Example:



]cd  \tmp

C:\Users\Danb\Desktop

will now switch to \tmp for the remaining of the session.

Command display

This command will display APL expressions using boxes around enclosed elements as per the familiar DISPLAY function.
Example:



]display ⍳¨⍳2 3
┌→────────────────────────────────────────────────┐

↓ ┌→──────┐ ┌→────────────┐ ┌→──────────────────┐ │

│ ↓ ┌→──┐ │ ↓ ┌→──┐ ┌→──┐ │ ↓ ┌→──┐ ┌→──┐ ┌→──┐ │ │

│ │ │1 1│ │ │ │1 1│ │1 2│ │ │ │1 1│ │1 2│ │1 3│ │ │

│ │ └~──┘ │ │ └~──┘ └~──┘ │ │ └~──┘ └~──┘ └~──┘ │ │

│ └∊──────┘ └∊────────────┘ └∊──────────────────┘ │
│ ┌→──────┐ ┌→────────────┐ ┌→──────────────────┐ │

│ ↓ ┌→──┐ │ ↓ ┌→──┐ ┌→──┐ │ ↓ ┌→──┐ ┌→──┐ ┌→──┐ │ │

│ │ │1 1│ │ │ │1 1│ │1 2│ │ │ │1 1│ │1 2│ │1 3│ │ │

│ │ └~──┘ │ │ └~──┘ └~──┘ │ │ └~──┘ └~──┘ └~──┘ │ │

│ │ ┌→──┐ │ │ ┌→──┐ ┌→──┐ │ │ ┌→──┐ ┌→──┐ ┌→──┐ │ │

│ │ │2 1│ │ │ │2 1│ │2 2│ │ │ │2 1│ │2 2│ │2 3│ │ │

│ │ └~──┘ │ │ └~──┘ └~──┘ │ │ └~──┘ └~──┘ └~──┘ │ │

│ └∊──────┘ └∊────────────┘ └∊──────────────────┘ │

└∊────────────────────────────────────────────────┘
Command fndiff

This command will show the different lines between 2 functions.
Example:



⎕fx'f1' '⍝f1' '123' '⍝x' 'ef1' 


      ⎕fx'f2' '⍝f2' '123' '⍝x' 'ef2'
      ]fndiff f1 f2

·f1 ·   ·   ·   ·   ·   ·   │·f2 ·   ·   ·   ·   ·   ·   

⍝f1 ·   ·   ·   ·   ·   ·   │⍝f2 ·   ·   ·   ·   ·   ·   

·ef1·   ·   ·   ·   ·   ·   │·ef2·   ·   ·   ·   ·   ·   

Command fnslike

This command will show all functions following a same pattern in their names. It accepts the –dos, -date and –format switches.
Example: display all functions containing the letter ‘a’
      ]fnsl a  -format
commandLineArgs disableSPICE   enableSPICE   regGetHandle                                          

disableSALT     enableSALT     qaEmptyCat    BootLocation                                                   

Command hex

This command will display a number in hexadecimal value or in decimal value if they happen to be in hex form already or if the –back switch is set.
Example:



]hex   FFF  A0

4095  160

Command latest

This command will list the names of the youngest functions changed (most likely today, otherwise of the last changed day), the most recently changed first.
Command nameslike

This command will show all objects following a same pattern in their names. Each name will be followed by the class of the name. It accepts the –dos, -date and –format switches.

Example:



]nameslike    a  -format

aplUtils.9        disableSALT.3     enableSALT.3 

commandLineArgs.2 disableSPICE.3    enableSPICE.3

Command pwd

This command will show you the current working directory. 
Command reordlocals

This command will reorder the local names in the header of the functions given in the argument. The argument is a series of patterns representing the names to be affected. It accepts the –dos, -date and –format switches.
Command sizeof

This command will show you the size of the variables and namespaces given in the argument. The argument is a series of patterns (including none) representing the names affected. It accepts the -top switch to limit the number of items shown.

Example:

      )obs

NStoScript      aplUtils        test


      )vars

CR      DELINS  Describe        FS



]size -top=4

NStoScript 132352

aplUtils    40964

test        31996

Describe    10128
Command varslike

This command will show all variables following a same pattern in their names. It accepts the –dos  and –format switches.
Command wsloc

This command will search strings in the current namespace. It accepts a number of switches that allow it to screen out hits in comments, text, etc. It accepts regular expressions and will perform replacement on most objects and APL files.
Group svn

This group contains a series of commands used as cover to SubVersions functions of the same name. For example, svnci commits changes made to the current working copy.
� this is not entirely true; what Spice really needs is []SALTUtils to be in []SE. When SALT is ON this is the case. But you may save your current session file while SALTUtils is there and then turn SALT OFF. Subsequent sessions will have SALTUtils present even though SALT has been turned OFF and Spice (user commands) will be available.


� ⎕SE.SALT.Save ‘timefns  Spice\timefns’  will do it


� This requires SALT/Spice version 1.3 or more. To see which version you are using type ⎕SE.SALT.Version


� UTC is sometimes denoted as Z time – Zero offset zone time – or Zulu time from the NATO phonetic alphabet


� The function does not deal with daylight savings time. An exercise for the reader?


� If you wish to delve into this subject have a look at Vector Vol 19, #4: Tools, Part 1. Basics.


� The value of date is ‘<90101’, the < included which is why the syntax includes BOTH = and <


� the 1st argument is surrounded by quotes because it contains a space





